Problem:
Problem Statement

AtCoder currently hosts three types of contests: ABC, ARC, and AGC. As the number of users has grown, in order to meet the needs of more users, AtCoder has decided to increase the number of contests to 26 types, from AAC to AZC. For convenience, we number these 26 types as type 1 through type 26. AtCoder wants to schedule contests for D days so that user satisfaction is as high as possible. For every day, AtCoder will hold exactly one contest, and each contest will end on that day. The satisfaction is calculated as follows.

The satisfaction at the beginning of day 1 is 0. Satisfaction can be negative.

Holding contests increases satisfaction. The amount of increase will vary depending on a variety of factors. Specifically, we know in advance that holding a contest of type i on day d will increase the satisfaction by s_{d,i}.

If a particular type of contest is not held for a while, the satisfaction decreases. Each contest type i has an integer c_i, and at the end of each day d=1,2,...,D, the satisfaction decreases as follows. Let \mathrm{last}(d,i) be the last day before day d (including d) on which a contest of type i was held. If contests of type i have never been held yet, we define \mathrm{last}(d,i)=0. At the end of day d, the satisfaction decreases by \sum _{i=1}^{26}c_i \times (d-\mathrm{last}(d,i)).

Please schedule contests on behalf of AtCoder.
If the satisfaction at the end of day D is S, you will get a score of \max(10^6 + S, 0).
There are 50 test cases, and the score of a submission is the total scores for each test case.
You can make submissions multiple times, and the highest score among your submissions will be your score.

Constraints

D = 365

Each c_i is an integer satisfying 0\leq c_i \leq 100.

Each s_{d,i} is an integer satisfying 0\leq s_{d,i} \leq 20000.

Input

Input is given from Standard Input in the following format:

D
c_1 c_2 \cdots c_{26}
s_{1,1} s_{1,2} \cdots s_{1,26}
\vdots
s_{D,1} s_{D,2} \cdots s_{D,26}

Output

Let t_d (1\leq t_d \leq 26) be the type of the contest that will be held at day d.
Print D integers t_d to Standard Output in the following format:

t_1
t_2
\vdots
t_D

Any output that does not follow the above format may result in 0 pointsWA for that test case.

Input Generation

Each integer c_i and s_{d,i} is generated independently and uniformly at random from the integers in the range described in the problem statement.

Sample Input 1

5
86 90 69 51 2 96 71 47 88 34 45 46 89 34 31 38 97 84 41 80 14 4 50 83 7 82
19771 12979 18912 10432 10544 12928 13403 3047 10527 9740 8100 92 2856 14730 1396 15905 6534 4650 11469 3628 8433 2994 10899 16396 18355 11424
6674 17707 13855 16407 12232 2886 11908 1705 5000 1537 10440 10711 4917 10770 17272 15364 19277 18094 3929 3705 7169 6159 18683 15410 9092 4570
6878 4239 19925 1799 375 9563 3445 5658 19857 11401 6997 6498 19933 3848 2426 2146 19745 16880 17773 18359 3921 14172 16730 11157 5439 256
8633 15862 15303 10749 18499 7792 10317 5901 9395 11433 3514 3959 5202 19850 19469 9790 5653 784 18500 10552 17975 16615 7852 197 8471 7452
19855 17918 7990 10572 4333 438 9140 9104 12622 4985 12319 4028 19922 12132 16259 17476 2976 547 19195 19830 16285 4806 4471 9457 2864 2192

Sample Output 1

1
17
13
14
13

Note that this example is a small one for checking the problem specification. It does not satisfy the constraint D=365 and is never actually given as a test case. The final satisfaction with this output is 79325, so the score is 1079325.

Input generator, score calculator, and visualizer

Beginner's Guide

If you don't know what to do, proceed to problem B or C.

Solution in Fortran (GNU Fortran 9.2.1) (judge verdict: Accepted)

module solver
    implicit none
    
contains
    !貪欲法による初期解の生成
    subroutine greedy(D,C,S,T,last,vmax)
        integer(8),intent(in)::D,C(26),S(:,:)
        integer(8),intent(out)::vmax
        integer(8),intent(inout)::T(:),last(:,:)
        integer(8),allocatable::last2(:),V(:),Tans(:),last_best(:,:)
        integer(8) dadd,i,i_calc,j,k,l,tbest,score,vbest
        
        allocate(Tans(D))
        allocate(last_best(26,D))
        allocate(last2(26))
        allocate(V(D))
        
        do dadd=0,26
            T=0
            V=0
            last=0
            last2=0
            !今日の日付+dadd日目のパラメータを使用して満足度を計算する
            do i=1,D
                i_calc=min(i+dadd,D)
                do j=1,26
                    !得点の初期化
                    if(i==1)V(i)=0
                    if(i>1)V(i)=V(i-1)
    
                    !lastの初期化
                    if(i==1)last(:,i)=0
                    if(i>1)last(:,i)=last(:,i-1)
    
                    last(j,i)=i
                    V(i)=V(i)+S(j,i)
                    
                    do k=1,26
                        do l=i,i_calc
                            V(i)=V(i)-C(k)*(l-last(k,i))
                        end do
                    end do
    
                    if(j==1)then
                        vmax=V(i)
                        tbest=j
                    else if(V(i)>=vmax)then
                        vmax=V(i)
                        tbest=j
                    end if
                end do
                if(i==1)last(:,i)=0
                if(i>1)last(:,i)=last(:,i-1)
                T(i)=tbest
                last(tbest,i)=i
            end do
    
            !得点の再計算
            score=0
            last2=0
            do i=1,D
                score=score+S(T(i),i)
                last2(T(i))=i
                do j=1,26
                    score=score-C(j)*(i-last2(j))
                end do
            end do
    
            if(dadd==0.or.score>vbest)then
                vbest=score
                Tans(:)=T(:)
                last_best(:,:)=last(:,:)
            end if
        end do

        vmax=vbest
        T(:)=Tans(:)
        last(:,:)=last_best(:,:)
        return
    end subroutine greedy

    subroutine change_one(D,C,S,T,V,last,vmax)
        integer(8),intent(in)::D,C(26),S(:,:)
        integer(8),intent(inout)::T(D),last(26,D),V(D),vmax
        integer(8)::drand,qrand,contests=26
        integer(8) torigin,i,j,last_sim(26,D),Vorigin(D)

        !乱数の生成
        drand=rnd_int(D)
        qrand=rnd_int(contests)

        last_sim(:,:)=last(:,:)
        Vorigin(:)=V(:)

        !drand日目に行うコンテストをqrandに変更する
        !last_simの書き換え
        torigin=T(drand)

        if(drand==1)then
            last_sim(T(drand),1)=0
            last_sim(qrand,1)=1
        else
            last_sim(T(drand),drand)=last(T(drand),drand-1)
            last_sim(qrand,drand)=drand
        end if

        !日付に対応するコンテストタイプの書き換え
        T(drand)=qrand

        !drand日目の得点の再計算
        if(drand==1)then
            V(drand)=0
        else
            V(drand)=V(drand-1)
        end if
        V(drand)=V(drand)+S(T(drand),drand)
        do i=1,26
            V(drand)=V(drand)-C(i)*(drand-last_sim(i,drand))
        end do

        !drand+1日目以降の得点の再計算
        do i=drand+1,D
            last_sim(:,i)=last_sim(:,i-1)
            V(i)=V(i-1)
            V(i)=V(i)+S(T(i),i)
            last_sim(T(i),i)=i
            do j=1,26
                V(i)=V(i)-C(j)*(i-last_sim(j,i))
            end do
        end do

        if(V(D)>vmax)then
            vmax=V(D)
            last(:,:)=last_sim(:,:)
        else
            V(:)=Vorigin(:)
            T(drand)=torigin
        end if
    end subroutine change_one

    !整数の乱数生成
    function rnd_int(rnd_max)
        integer(8),intent(in)::rnd_max
        integer(8)::rnd_int
        real::rnd_real
        call random_number(rnd_real)
        rnd_int=mod(int(rnd_real*100000000),rnd_max)+1
        return
    end function
end module solver

program scoring
    use solver
    implicit none
    integer(8)::t1,t2,t_rate,t_max,diff!実行時間の計測関連
    integer(8)::D,C(26),last(26,365),last_sim(26,365),vmax,i
    integer(8)::steps=0!ベンチマーク用
    integer(8),allocatable::S(:,:),T(:),Tans(:),V(:),Vorigin(:)
    !時間計測開始
    call system_clock(t1)

    !入力
    read*,D,C
    allocate(S(26,D))
    allocate(T(D))
    allocate(Tans(D))
    allocate(V(D))
    allocate(Vorigin(D))
    read*,S
    T=0
    V=0
    last=0
    last_sim=0

    !貪欲法による初期解の生成
    call greedy(D,C,S,T,last,vmax)
    call system_clock(t2,t_rate,t_max)
    diff=t2-t1

    do
        if(diff>1990*t_rate/1000)exit
        steps=steps+1
        
        call change_one(D,C,S,T,V,last,vmax)

        !時間計測
        call system_clock(t2,t_rate,t_max)
        diff=t2-t1
    end do
    !print*,diff/dble(t_rate),steps

    !答えの出力
    do i=1,D
        print'(i0)',T(i)
    end do
contains

end program scoring